


Challenge October 2024 Cheryl's Birthday



A solution with IBM Watsonx AI CPLEX by Alex Fleischer
afleischer@fr.ibm.com

CPLEX is available within IBM AI platform watsonx.AI

I relied on OPL, the high level modeling language within CPLEX. I modeled this as 3 enumerations with Constraint Programming (CP Optimizer) within CPLEX

I used the model .mod:

```
using CP;

int step=...; // 1,2 or 3

assert step in 1..3;

tuple date
{
    int m;
    int d;
}

// To get a nicer display
string monthNames[1..12]=
["January","February","March","April","May","June",
"July","August","September","October","November","December"];
```

```

// Lines for options.dat
// step=1;
// options={<5,15>,<5,16>,<5,19>,<6,17>,
//<6,18>,<7,14>,<7,16>,<8,14>,<8,15>,<8,17>};
// End of lines in options.dat

// Allowed dates
{date} options=...;
// The new options we compute (a subset of previous options)
{date} newoptions={};

sorted {int} monthoptions={o.m | o in options};
sorted {int} dayoptions={o.d | o in options};

int isAnOption[m in monthoptions][d in dayoptions]=(<m,d>) in options;

assert sum(m in monthoptions,d in dayoptions) isAnOption[m][d]==card(options);

int nbdayoptionsforthismonth[m in monthoptions]=card({o | o in options : o.m==m});
int nbmonthoptionsforthisday[d in dayoptions]=card({o | o in options : o.d==d});

execute
{
    nbdayoptionsforthismonth;
}
dvar int month;
dvar int day;

subject to
{

    allowedAssignments(options,month,day);

// Cheryl then tells Albert and Bernard separately the month and the day
// of her birthday respectively.

//Albert: I don't know when Cheryl's birthday is, but I know that Bernard doesn't
know too.

if (step==1)
{
    nbdayoptionsforthismonth[month]>=2;
    forall(d in dayoptions)
        (isAnOption[month][d]==1) => (nbmonthoptionsforthisday[d]>=2);
}

//Bernard: At first I didn't know when Cheryl's birthday is, but I know now.

if (step==2) nbmonthoptionsforthisday[day]==1;

//Albert: Then I also know when Cheryl's birthday is.

if (step==3) nbdayoptionsforthismonth[month]==1;

}

```

```

// Display result
execute
{
    writeln(monthNames[month], " ", day);
}

// Loop to enumerate all solutions
main
{
    writeln("Step = ", thisOplModel.step);
    var nbsol=0;
    // Depth search and one worker only
    cp.param.SearchType=24;
    cp.param.workers=1;

    thisOplModel.generate();
    cp.startNewSearch();
    while
    (cp.next()) { thisOplModel.postProcess(); nbsol++;
    thisOplModel.newoptions.add(thisOplModel.month, thisOplModel.day); }
    writeln("nb solutions=", nbsol);

    // Write new options back in step.dat and increment step
    var output=new IloOplOutputFile("step.dat");
    output.writeln("step=", thisOplModel.step+1);
    output.writeln(";");
    output.writeln("options=");
    output.writeln(thisOplModel.newoptions);
    output.writeln(";");
    output.close();
}

```

And the data .dat

```

step=1;
options={<5,15>,<5,16>,<5,19>,<6,17>,
<6,18>,<7,14>,<7,16>,<8,14>,<8,15>,<8,17>};

```

And that gives if I run this 3 times

```

Step = 1
July 14
July 16

```

August 14
August 15
August 17
nb solutions=5

Step = 2
July 16
August 15
August 17
nb solutions=3

Step = 3
July 16
nb solutions=1

Cheryl was born July 16th